```
(*Edgeworth expansion for Stirling numbers of the first kind
 We use the expansion and the notation from the paper
 "General Edgeworth expansions with applications to profiles of random trees"
 by Zakhar Kabluchko, Alexander Marynych, Henning Sulzbach
  available at http://arxiv.org/abs/1606.03920
*)
```

```
In[4]:= (*Parameters:
  phi(beta) = Exp[beta]-1,
  LogW[beta] = LogGamma[theta]-LogGamma[theta*Exp[beta]],
  kappa[[j]] = j-th derivative of phi(beta)=Exp[beta]-1  at beta=0,
  chi[[j]] = j-th derivative of LogW at beta=0,
  sigma[[j]] = 1 is ignored everywhere.
 *)
kappa = Table[1, {n, 1, 5}]
LogW[beta_] = LogGamma[theta] - LogGamma[theta * Exp[beta]]
chi = List[D[LogW[x], x] /. x → 0,
  D[D[LogW[x], x], x] /. x → 0, D[D[D[LogW[x], x], x], x] /. x → 0]
```

Out[4]= {1, 1, 1, 1, 1}

Out[5]= LogGamma[theta] − LogGamma[e^{beta} theta]

Out[6]= {−theta PolyGamma[0, theta],
 −theta PolyGamma[0, theta] − $theta^2$ PolyGamma[1, theta],
 −theta PolyGamma[0, theta] −
  3 $theta^2$ PolyGamma[1, theta] − $theta^3$ PolyGamma[2, theta]}

```
(*Differential operators D_1,D_2,D_3. Here D stays for sigma^(-1) d/dx *)
```

```
In[7]:= D1 = kappa[[3]] / 6 * D^3 + chi[[1]] * D^1
D2 = kappa[[4]] / 12 * D^4 + chi[[2]] * D^2
D3 = kappa[[5]] / 20 * D^5 + chi[[3]] * D^3
```

Out[7]= $\frac{D^3}{6}$ − D theta PolyGamma[0, theta]

Out[8]= $\frac{D^4}{12}$ + D^2 (−theta PolyGamma[0, theta] − $theta^2$ PolyGamma[1, theta])

Out[9]= $\frac{D^5}{20}$ + D^3 (−theta PolyGamma[0, theta] −
 3 $theta^2$ PolyGamma[1, theta] − $theta^3$ PolyGamma[2, theta])

In[10]:=
```
(*
Bell polynomials are
 B_0=1,
 B_1(z_1) = z_1,
 B_2(z_1,z_2) = z_1^2 + z_2,
 B_3(z_1,z_2,z_3) = z_1^3 + 3z_1z_2 + z_3
*)
(*
Probabilist Hermite polynomials are given by
  He_p(x) = 2^(-p/2)*HermiteH[p, x/Sqrt[2]]
  To check this,
use: Table[Simplify[2^(-p/2)*HermiteH[p, x/Sqrt[2]]], {p,0,6}]
*)

(*Terms in the Edgeworth expansion of the
 Stirling numbers are denoted by G_0(x), G_1(x),...*)
H0 = 1;
```

In[11]:=
```
H1[x_] = Simplify[ (Expand[D1] /. D^p_ -> 2^(-p / 2) * HermiteH[p, x / Sqrt[2]]) /.
   D -> 2^(-1 / 2) * HermiteH[1, x / Sqrt[2]]]
```

Out[11]= $\frac{1}{6} x \left(-3 + x^2 - 6 \text{ theta PolyGamma}[0, \text{theta}]\right)$

In[27]:=
```
A11 = Simplify[Coefficient[H1[x], x]]
A12 = Simplify[Coefficient[H1[x], x^3]]
```

Out[27]= $-\frac{1}{2} - \text{theta PolyGamma}[0, \text{theta}]$

Out[28]= $\frac{1}{6}$

In[12]:=
```
H2[x_] = Simplify[
  1 / (2!) * Expand[D1^2 + D2] /. D^p_ -> 2^(-p / 2) * HermiteH[p, x / Sqrt[2]]]
```

Out[12]= $\frac{1}{72} \left(-6 + 27 x^2 - 12 x^4 + x^6 - 12 \text{ theta } x^2 \left(-3 + x^2\right) \text{PolyGamma}[0, \text{theta}] + 36 \text{ theta}^2 \left(-1 + x^2\right) \text{PolyGamma}[0, \text{theta}]^2 - 36 \text{ theta}^2 \left(-1 + x^2\right) \text{PolyGamma}[1, \text{theta}]\right)$

In[32]:=
```
A21 = Simplify[H2[0]]
A22 = Simplify[Coefficient[H2[x], x^2]]
```

Out[32]= $\frac{1}{12} \left(-1 - 6 \text{ theta}^2 \text{PolyGamma}[0, \text{theta}]^2 + 6 \text{ theta}^2 \text{PolyGamma}[1, \text{theta}]\right)$

Out[33]= $\frac{1}{8} \left(3 + 4 \text{ theta PolyGamma}[0, \text{theta}] + 4 \text{ theta}^2 \text{PolyGamma}[0, \text{theta}]^2 - 4 \text{ theta}^2 \text{PolyGamma}[1, \text{theta}]\right)$

In[15]:=
```
H3[x_] = Simplify[1 / (3 !) * Expand[D1^3 + 3 D1 * D2 + D3] /.
    D^p_ -> 2^(-p / 2) * HermiteH[p, x / Sqrt[2]]]
```

Out[15]=
$$\frac{1}{6480} x \left(810 - 2115 x^2 + 999 x^4 - 135 x^6 + 5 x^8 + 540 \text{theta}^2 \left(-3 - 4 x^2 + x^4\right) \text{PolyGamma}[0, \text{theta}]^2 - 1080 \text{theta}^3 \left(-3 + x^2\right) \text{PolyGamma}[0, \text{theta}]^3 - 540 \text{theta}^2 \left(-3 - 4 x^2 + x^4\right) \text{PolyGamma}[1, \text{theta}] + 90 \text{theta} \,\text{PolyGamma}[0, \text{theta}] \left(6 - 27 x^2 + 12 x^4 - x^6 + 36 \text{theta}^2 \left(-3 + x^2\right) \text{PolyGamma}[1, \text{theta}]\right) + 3240 \text{theta}^3 \text{PolyGamma}[2, \text{theta}] - 1080 \text{theta}^3 x^2 \text{PolyGamma}[2, \text{theta}]\right)$$

In[34]:=
```
A31 = Simplify[Coefficient[H3[x], x]]
```

Out[34]=
$$\frac{1}{24} \left(3 - 6 \text{theta}^2 \text{PolyGamma}[0, \text{theta}]^2 + 12 \text{theta}^3 \text{PolyGamma}[0, \text{theta}]^3 + 6 \text{theta}^2 \text{PolyGamma}[1, \text{theta}] + \text{PolyGamma}[0, \text{theta}] \left(2 \text{theta} - 36 \text{theta}^3 \text{PolyGamma}[1, \text{theta}]\right) + 12 \text{theta}^3 \text{PolyGamma}[2, \text{theta}]\right)$$

```
(*
The first three terms in the expansion for x = a/Sqrt[w].
     The term H4/w^2 = constant/w^2 +
   o(1/w^2) is ignored because the constant does not depend on a.
     Error: o(1/w^2)
*)
Series[Exp[-t^2 / 2], {t, 0, 4}]
```

Out[14]=
$$1 - \frac{t^2}{2} + \frac{t^4}{8} + O[t]^5$$

In[16]:=

```
A = (1 + H1[a / Sqrt[w]] / Sqrt[w] + H2[a / Sqrt[w]] / w + H3[a / Sqrt[w]] / w^(3 / 2)) *
   (1 - a^2 / (2 w) + 1 / 8 * a^4 / w^2);
Collect[Coefficient[A, 1 / w], a]
Collect[Coefficient[A, 1 / w^(1 / 2)], a]
Collect[Coefficient[A, 1 / w^(3 / 2)], a]
Collect[Coefficient[A, 1 / w^2], a]
```

Out[17]= $-\frac{1}{12}-\frac{a^2}{2}-\frac{1}{2}\text{theta}^2\,\text{PolyGamma}[0,\text{theta}]^2+a\left(-\frac{1}{2}-\text{theta PolyGamma}[0,\text{theta}]\right)+\frac{1}{2}\text{theta}^2\,\text{PolyGamma}[1,\text{theta}]$

Out[18]= 0

Out[19]= 0

Out[20]= $\frac{a^4}{8}+a^3\left(\frac{5}{12}+\frac{1}{2}\text{theta PolyGamma}[0,\text{theta}]\right)+a^2\left(\frac{5}{12}+\frac{1}{2}\text{theta PolyGamma}[0,\text{theta}]+\frac{3}{4}\text{theta}^2\,\text{PolyGamma}[0,\text{theta}]^2-\frac{3}{4}\text{theta}^2\,\text{PolyGamma}[1,\text{theta}]\right)+a\left(\frac{1}{8}+\frac{1}{12}\text{theta PolyGamma}[0,\text{theta}]-\frac{1}{4}\text{theta}^2\,\text{PolyGamma}[0,\text{theta}]^2+\frac{1}{2}\text{theta}^3\,\text{PolyGamma}[0,\text{theta}]^3+\frac{1}{4}\text{theta}^2\,\text{PolyGamma}[1,\text{theta}]-\frac{3}{2}\text{theta}^3\,\text{PolyGamma}[0,\text{theta}]\,\text{PolyGamma}[1,\text{theta}]+\frac{1}{2}\text{theta}^3\,\text{PolyGamma}[2,\text{theta}]\right)$

In[21]:=

```
(*Here we insert a = gam + chi[[1]]-1/2, where gam is the new variable*)
Poly[gam_] = Simplify[Coefficient[A, 1 / w^2] /. a → gam + chi[[1]] - 1 / 2]
```

Out[21]= $\frac{1}{384}(-1+2\,\text{gam}-2\,\text{theta PolyGamma}[0,\text{theta}])\left(1+18\,\text{gam}+44\,\text{gam}^2+24\,\text{gam}^3+4\,(-19+6\,\text{gam})\,\text{theta}^2\,\text{PolyGamma}[0,\text{theta}]^2+24\,\text{theta}^3\,\text{PolyGamma}[0,\text{theta}]^3-24\,(-5+6\,\text{gam})\,\text{theta}^2\,\text{PolyGamma}[1,\text{theta}]-2\,\text{theta PolyGamma}[0,\text{theta}]\left(13+44\,\text{gam}-12\,\text{gam}^2+72\,\text{theta}^2\,\text{PolyGamma}[1,\text{theta}]\right)+96\,\text{theta}^3\,\text{PolyGamma}[2,\text{theta}]\right)$

In[22]:=

```
(*
Now we can compute the quantity denoted by s^*
 (theta) in the paper "Asymptotic expansions for the Ewens
   distribution and the Stirling numbers of the first kind"
*)
Simplify[Poly[1 / 2] - Poly[-1 / 2]]
```

Out[22]= $\frac{1}{2}\text{theta}^2\,(2\,\text{PolyGamma}[1,\text{theta}]+\text{theta PolyGamma}[2,\text{theta}])$

In[24]:=
```
(*The resulting expression is positive*)
Plot[Poly[1 / 2] - Poly[-1 / 2], {theta, 0, 1}]
```

Out[24]=